1 Mapleの線形代数

線形代数に関連するコマンドです．ベクトル，行列などは array や list を工夫しても使えますが，LinearAlgebra パッケージが用意してくれているデータ構造や関数を使うと，高速・簡単に線形代数計算が出来ます．厄介な逆行列や固有値も行列を生成さえすればすぐに求まります．

1.1 ベクトル，行列の生成

先ず

```
> with(LinearAlgebra):
```

が必要です．次に Matrix, Vector の生成方法ですが，以下のようにいくつもの方法が用意されています．

先ずは標準的な Vector コマンドを使ったベクトルの生成．

```
> v1:=Vector([x,y,z]);
```

$$v1 := \begin{bmatrix} x \\ y \\ z \end{bmatrix}$$

ここで Vector は縦 (列) ベクトル (column vector) を生成することに注意ください．横 (行) ベクトル (row vector) の生成は以下の通りです．(英語で座席は row, 新聞の囲み記事は column です)

```
> v2:=Vector[row]([x,y,z]);
```

$$v2 := \begin{bmatrix} x & y & z \end{bmatrix}$$

{ }や [ ] のかわりに，< >と|を使っても縦・横ベクトルが生成できます．

```
> v2:=<1,2,3>;
```

$$v2 := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

```
> v3:=<1|2|3>;
```

$$v3 := \begin{bmatrix} 1 & 2 & 3 \end{bmatrix}$$

次は標準的な Matrix コマンドを使った行列の生成．2 行 3 列の行列を listlist から生成しています．

```
> A0:=Matrix(2,3,[[1,2,3],[4,5,6]]);
```

$$A0 := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

ベクトルと同様に`< >`と`|`を使っての生成です.

```
> A1:=<<1,2,3>|<4,5,6>|<7,8,9>>;
```

$$A1 := \begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 9 \end{bmatrix}$$

特殊な形状を指定する事が出来ます．ここでは単位行列 (shape=identity) を生成しています.

```
> E:=Matrix(3,3,shape=identity);
```

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

listlist から convert を使った変換です.

```
> A3:=[[1,2],[3,4]];
> A4:=convert(A3,Matrix);
```

$$A3 := [[1, 2], [3, 4]]$$

$$A4 := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

1.2 ベクトル，行列の演算

先ずは和とスカラー積です．通常の数値，変数の算術演算と同様の記法です.

```
> A5:=Matrix(2,2,[[3,-1],[1,2]]):
> a*A4+b*A5;
```

$$\begin{bmatrix} a + 3\,b & 2\,a - b \\ 3\,a + b & 4\,a + 2\,b \end{bmatrix}$$

"."(ピリオド) が行列の積を表わします.

```
> A4.A4;
```

$$\begin{bmatrix} 7 & 10 \\ 15 & 22 \end{bmatrix}$$

"."は行列とベクトル，あるいはベクトル同士の内積にも使われます.

```
> A1.v1;
```

$$\begin{bmatrix} x + 4\,y + 7\,z \\ 2\,x + 5\,y + 8\,z \\ 3\,x + 6\,y + 9\,z \end{bmatrix}$$

```
> v2.v1;
```

$$x+2y+3z$$

行列の積で次元があわないときには `Error` が返ってきます.

```
> v1.A1;
```

```
Error, (in LinearAlgebra:-VectorMatrixMultiply) invalid input:
‘LinearAlgebra:-VectorMatrixMultiply‘ expects its 1st argument, v,
to be of type Vector[row] but received Vector[Column]( 1..3,[ x, y, z] ,
datatype = anything, storage = rectangular, order = C_order )}
```

ベクトルの外積 (outer product) は `OuterProductMatrix` です.

```
> OuterProductMatrix(v1,v2);
```

$$\begin{bmatrix} x & 2x & 3x \\ y & 2y & 3y \\ z & 2z & 3z \end{bmatrix}$$

1.3 逆行列，行列式，転置

逆行列は `MatrixInverse` で一発で求まります.

```
> A3:=Matrix(3,3,[[1,2,1],[4,5,6],[7,8,9]]);
> MatrixInverse(A3);
```

$$A3 := \begin{bmatrix} 1 & 2 & 1 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

$$\begin{bmatrix} -1/2 & -5/3 & 7/6 \\ 1 & 1/3 & -1/3 \\ -1/2 & 1 & -1/2 \end{bmatrix}$$

行列式は，`Determinant` です.

```
> Determinant(A3);
```

$$6$$

`A1` の `MatrixInverse` を試みると

```
> MatrixInverse(A1);
```

```
Error, (in LinearAlgebra:-LA_Main:-MatrixInverse) singular matrix
```

と叱られます．これは行列式が 0 だからです.

```
> Determinant(A1);
```

$$0$$

転置 `Transpose` の使用例です.

```
> Transpose(A0);
```

$$\begin{bmatrix} 1 & 4 \\ 2 & 5 \\ 3 & 6 \end{bmatrix}$$

```
> Transpose(v1);
```

$$\begin{bmatrix} x & y & z \end{bmatrix}$$

```
> Transpose(v1).v1;
```

$$x^2 + y^2 + z^2$$

1.4 固有値

固有値，固有ベクトルは `Eigenvectors` で一発で求まります．

```
> Eigenvectors(A1);
```

$$\begin{bmatrix} 0 \\ 15/2 + 3/2\sqrt{33} \\ 15/2 - 3/2\sqrt{33} \end{bmatrix}, \begin{bmatrix} 1 & 4/3\,\frac{\frac{99}{2}+21/2\sqrt{33}}{\left(11/2+3/2\sqrt{33}\right)\left(13/2+3/2\sqrt{33}\right)} & 4/3\,\frac{\frac{99}{2}-21/2\sqrt{33}}{\left(11/2-3/2\sqrt{33}\right)\left(13/2-3/2\sqrt{33}\right)} \\ -2 & 1/12\,\frac{\frac{165}{2}+21/2\sqrt{33}}{11/2+3/2\sqrt{33}} & 1/12\,\frac{\frac{165}{2}-21/2\sqrt{33}}{11/2-3/2\sqrt{33}} \\ 1 & 1 & 1 \end{bmatrix}$$

浮動小数点に直します．l(lambda) に固有値を，V に固有ベクトルを格納します．

```
> l,V:=evalf(Eigenvectors(A1));
```

$$l, V := \begin{bmatrix} 16.11684397 \\ -1.116843970 \\ 0.0 \end{bmatrix}, \begin{bmatrix} 0.6861406616 & -2.186140661 & 1.0 \\ 0.8430703308 & -0.5930703307 & -2.0 \\ 1.0 & 1.0 & 1.0 \end{bmatrix}$$

V の列ベクトルが対応する固有ベクトルです．`Column` で行列の行を要素とするベクトルが作れます．これを用いて固有値方程式

$$A1.V_2 = \lambda_2 V_2 \tag{1}$$

を確かめてみます．

```
> l[2].Column(V,2);
```

$$\begin{bmatrix} 2.44157801480966396 \\ 0.662367022628200908 \\ -1.11684396999999991 \end{bmatrix}$$

```
> A1.Column(V,2);
```

$$\begin{bmatrix} 2.44157801619999936 \\ 0.662367024499999957 \\ -1.11684396720000123 \end{bmatrix}$$

ついでに行ベクトルの取り出しは `Row` です.

1.5 固有値の幾何学的表示

次章で解説する予定のいくつかの plots 関数を使って 2 次元正方行列の幾何学的な意味を示します. 先ずは必要なパッケージを読み込み, 単純に `Eigenvectors` で固有値と固有ベクトルを求めます.

```
> restart;
> with(LinearAlgebra):with(plots):with(plottools):
> A:=Matrix(1..2,1..2,[[3,2/3],[2/3,2]]);
```

$$A := \begin{bmatrix} 3 & 2/3 \\ 2/3 & 2 \end{bmatrix}$$

```
> (l,V):=Eigenvectors(A);
```

$$l, V := \begin{bmatrix} 10/3 \\ 5/3 \end{bmatrix}, \begin{bmatrix} 2 & -1/2 \\ 1 & 1 \end{bmatrix}$$

固有ベクトルの長さが 1 となるように規格化します.

```
> V1:=Normalize(Column(V,1),Euclidean);
> V2:=Normalize(Column(V,2),Euclidean);
```

$$V1 := \begin{bmatrix} 2/5\sqrt{5} \\ 1/5\sqrt{5} \end{bmatrix}$$

$$V2 := \begin{bmatrix} -1/5\sqrt{5} \\ 2/5\sqrt{5} \end{bmatrix}$$

規格化したベクトルで作った直交行列を U として, $U^T AU$ により対角化されているか確認します.

```
> V:=Matrix([V1,V2]);
```

$$U := \begin{bmatrix} 2/5\sqrt{5} & -1/5\sqrt{5} \\ 1/5\sqrt{5} & 2/5\sqrt{5} \end{bmatrix}$$

```
> Transpose(U).A.U;
```

$$\begin{bmatrix} 10/3 & 0 \\ 0 & 5/3 \end{bmatrix}$$

次に行列 A による座標変換の様子を見ます. 以下のスクリプトでは `x0` で作った円上の等間隔の点が `x1:=A.x0` で変換されます. それぞれの点を pointplot し,

それぞれの変換前後を line コマンドで線で結んでいます.

```
>  N:=30:p1:=[]:l1:=[]:
>  for k from 0 to N-1 do
>  x0:=Vector([sin(2*Pi*k/N),cos(2*Pi*k/N)]);
>  x1:=A.x0;
>  p1:=[op(p1),pointplot(x0,x1)];
>  l1:=[op(l1),line( evalf(convert(x0,list)),evalf(convert(x1,list))
)];
>  end do:
```

固有ベクトルで表される主軸の傾きを $slope = \Delta y/\Delta x = V[2]/V[1]$ で求めています. この傾きを持った二つの直線を plot し principal に入れています.

```
>  slope1:=V1[2]/V1[1];
>  slope2:=V2[2]/V2[1];
```

$$slope1 := 1/2$$
$$slope2 := -2$$

```
>  principal:=plot(slope1*x,slope2*x,x=-4..4):
```

これら全てを一度に表示します.

```
>  display([principal,op(p1),op(l1)],view=[-4..4,-4..4],axes=box);
```



上図から対称行列に対する固有ベクトル (あるいは主軸) の重要な振る舞いが直観

的に読み取れます.

1. 主軸上にある点 (固有ベクトル) は，まさに固有値方程式

$$Ax = \lambda x \tag{2}$$

を満たしていることが期待できます．つまり，主軸上にある点は行列変換によっても主軸から外れずに (定数倍しただけ) 移動します．また，

2. 主軸は直交しています.

演習問題

1. **逆行列**
次の連立方程式の解を求める.

```
2x + 3y -z = -3
2x +  y -z =  1
 x + 3y +z = -6
```

(a) 左辺の係数で行列 A をつくる.
(b) その逆行列 A^{-1} を求める.
(c) $A^{-1}A = E$ を確認せよ.
(d) 右辺の値で作ったベクトル b と逆行列を掛け,

$$\begin{aligned} Ax &= b \\ A^{-1}Ax &= A^{-1}b \\ Ex &= A^{-1}b \end{aligned} \tag{3}$$

より解を求める.

2. **固有値**
次の対称行列

```
H:=Matrix(2,2,[[1,1],[1,3]]);
```

の固有値を Eigenvalues を使って求めよ．また，

```
H2:= H - x * Matrix(2,2,shape=identity);
```

の行列式から 2 次方程式をつくり，その解を solve を使って求めた結果と比較せよ.